Overview

The GREMOVE procedure processes a map data set that is used as input. It does not produce any graphics output. Instead, it produces an output data set that typically becomes the input map data set for the GMAP procedure (see Chapter 19, “The GMAP Procedure,” on page 731).

The GREMOVE procedure combines unit areas defined in a map data set into larger unit areas by removing shared borders between the original unit areas. For example, Figure 25.1 on page 903 and Figure 25.2 on page 904 show combined unit areas in a typical map data set by removing state boundaries to create regional census divisions.

Figure 25.1  Map before Removing Borders (GR25N01(a))
The program for these maps is shown in Example 1 on page 908.

Concepts

The GREMOVE procedure processes the input map data set to remove internal boundaries and creates a new map data set. The PROC GREMOVE statement identifies the input and output map data sets. The ID statement identifies the variable or variables in the input map data set that define the current unit areas. The BY statement identifies the variable or variables in the input map data set that define the new unit areas.

About the Input Map Data Set

The input map data set must be in map data set format (see “About Map Data Sets” on page 735) and it must contain these variables:

- a numeric variable named X that contains the horizontal coordinates of the map boundary points.
- a numeric variable named Y that contains the vertical coordinates of the map boundary points.
- one or more variables that uniquely identify the current unit areas in the map. These variables are listed in the ID statement. Each group of observations with a different ID variable value is evaluated as a separate unit area.
- one or more variables that identify the new unit areas to be created in the output map data set. These variables are listed in the BY statement.

It may also contain the variable SEGMENT, which is used to distinguish non-contiguous segments of the same unit areas. Other variables may exist in the input map data set, but they do not affect the GREMOVE procedure and they will not be carried into the output map data set.

About the Output Map Data Set

The output map data set contains the newly defined unit areas. These new unit areas are created by removing all interior line segments from the original unit areas. All variables in the input map data set except X, Y, SEGMENT, and the variables listed in the BY statement are omitted from the output map data set.
The output map data set may contain missing X, Y coordinates to construct any polygons that have enclosed boundaries (like lakes or combined regions that have one or more hollow interior regions).

The SEGMENT variable in the output map data set is ordered according to the size of the bounding box around the polygon that it describes. A SEGMENT value of 1 describes the polygon whose bounding box is the largest in the unit area and so on. This information is useful for removing small polygons that clutter up maps.

All current unit areas with common BY-variable value(s) are combined into a single unit area in the output map data set. The new unit area contains

- all boundaries that are not shared, such as islands and lakes
- all boundaries that are shared by two different BY groups.

### About Unmatched Area Boundaries

If you are using map data sets in which area boundaries do not match precisely (for example, if the boundaries were digitized with a different set of points), PROC GREMOVE will not be able to identify common boundaries properly, resulting in abnormalities in your output data set.

If the points in the area boundaries match up except for precision differences, before using PROC GREMOVE round each X and Y value in your map data set accordingly, using the DATA step function ROUND. See SAS Language Reference: Dictionary for information on the ROUND function.

For example, if you have a map data set named APPROX in which the horizontal and vertical coordinate values for interior boundaries of unit areas are exactly equal only to three decimal places, this DATA step creates a new map data set, EXACT, that is better suited for use with the GREMOVE procedure:

```sas
data exact;
  set approx;
  if x ne . then x=round(x,.001);
  if y ne . then y=round(y,.001);
run;
```

### Procedure Syntax

**Requirements:** The BY and ID statements are required.

```sas
PROC GREMOVE <DATA=input-map-data-set>
  <OUT=output-map-data-set>
  BY <DESCENDING>variable1
  <...<DESCENDING>variableN
  <NOTSORTED>
  ID variable(s);
```
**PROC GREMOVE Statement**

Identifies the input and output map data sets.

*Requirements:* An input map data set is required.

**Syntax**

```plaintext
PROC GREMOVE <DATA=input-map-data-set>
   <OUT=output-map-data-set>;
```

**Options**

**DATA=input-map-data-set**

specifies the map data set that is to be processed. By default, the procedure uses the most recently created SAS data set. The GREMOVE procedure expects the observations in the input map data set to be sorted in ascending order of the BY-variable values.

*See also:* “About the Input Map Data Set” on page 904 and “SAS Data Sets” on page 25

*Featured in:* Example 2 on page 912

**OUT=output-data-set**

names the new map data set, which contains the coordinates of the new unit areas created by the GREMOVE procedure. By default, the GREMOVE procedure names the new data set using the DATAn naming convention. That is, the procedure uses the name WORK.DATAn, where n is the next unused number in sequence. Thus, the first automatically named data set is DATA1, the second is DATA2, and so on.

*See also:* “About the Output Map Data Set” on page 904

*Featured in:* Example 2 on page 912

**BY Statement**

Lists the variable or variables that identify the new unit areas.

*Requirements:* At least one variable is required.

*See also:* “BY Statement” on page 172

*Featured in:* Example 1 on page 908

**Syntax**

```plaintext
BY <DESCENDING>variable1
   <...<DESCENDING>variable-n>
   <NOTSORTED>;
```
Required Arguments

**variable(s)**
identifies one or more variables in the input map data set that define the new unit areas. Variable(s) can be either numeric or character.

The BY variables in the input map data set become the ID variables for the output map data set.

Options

**DESCENDING**
indicates that the input map data set is sorted in descending order. By default, the GREMOVE procedure expects all BY-variable values to appear in ascending order.

This option affects only the variable that immediately follows the option.

**NOTSORTED**
indicates that observations with the same BY-variable values are to be grouped as they are encountered without regard for whether the values are in alphabetical or numerical order. NOTSORTED can appear anywhere in the BY statement. It affects all of the variables that are specified in the statement. NOTSORTED overrides DESCENDING if both appear in the same BY statement.

Ordering Observations

To sort the input map data set, use the SORT procedure in base SAS, for example

/* arrange the observations in desired order */
proc sort data=mapdata out=mapsort;
    by state;
run;
/* remove the county boundaries */
proc gremove data=mapsort out=newmap;
    by state;
    id county;
run;

Notice that the GREMOVE procedure uses the same BY statement as the SORT procedure.
See the SAS Procedures Guide for further information on the SORT procedure.

Note: If an observation is encountered for which the BY-variable value is out of the proper order, the GREMOVE procedure stops and issues an error message.

ID Statement

Identifies the variable or variables that define the hierarchy of the current unit areas in the input map data set.

*Requirements:* At least one id-variable is required.

*Featured in:* Example 1 on page 908
Syntax

ID id-variable(s);

Required Arguments

id-variable(s)
specifies one or more variables in the input map data set that identify the unit areas to be combined. These variables are not included in the output map data set. Id-variable(s) can be either numeric or character.

See also: “About the Input Map Data Set” on page 904

Examples

The following examples illustrate major features of the GREMOVE procedure.

Example 1: Removing State Boundaries from U.S. Map

Procedure features:
  BY statement
  ID statement

Other features:
  SORT procedure
  MERGE procedure
  LIBNAME statement

Sample library member: GR25N01

This example processes the MAPS.US map data set, supplied with SAS/GRAPH, to produce a new map data set containing boundaries for the U.S. Bureau of the Census divisions. Because the MAPS.US map data set does not contain a variable to identify any unit area other than states, this example creates a map data set that contains the census divisions and that can be processed with the GREMOVE procedure.

The STATE variable in the MAPS.US data set, containing numeric FIPS codes for each state, is used as the BY-variable to merge the REFLIB.CBSTATES and MAPS.US data sets. Output 25.1 on page 909 shows the variables that are present in the data set before using the GREMOVE procedure:
Output 25.1  The MAPS.US Data Set

<table>
<thead>
<tr>
<th>OBS</th>
<th>STATE</th>
<th>SEGMENT</th>
<th>X</th>
<th>Y</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>1</td>
<td>0.16175</td>
<td>-0.10044</td>
</tr>
<tr>
<td>2</td>
<td>1</td>
<td>1</td>
<td>0.12305</td>
<td>-0.10415</td>
</tr>
<tr>
<td>3</td>
<td>1</td>
<td>1</td>
<td>0.12296</td>
<td>-0.10678</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1524</td>
<td>56</td>
<td>1</td>
<td>-0.18757</td>
<td>0.15035</td>
</tr>
<tr>
<td>1525</td>
<td>56</td>
<td>1</td>
<td>-0.10158</td>
<td>0.13997</td>
</tr>
<tr>
<td>1526</td>
<td>56</td>
<td>1</td>
<td>-0.10398</td>
<td>0.11342</td>
</tr>
</tbody>
</table>

And Figure 25.3 on page 909 shows the map before processing:

Figure 25.3  Map before Removing Borders (GR25N01(a))

Output 25.2 on page 910 shows the variables that are present in the data set after you use the GREMOVE procedure. Notice that the new map data set contains a new variable called DIVISION:
Example 1: Removing State Boundaries from U.S. Map

Chapter 25

Output 25.2  The REMSTATE Data Set

<table>
<thead>
<tr>
<th>OBS</th>
<th>X</th>
<th>Y</th>
<th>SEGMENT</th>
<th>DIVISION</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>0.29825</td>
<td>0.17418</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>2</td>
<td>0.29814</td>
<td>0.17820</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>3</td>
<td>0.30206</td>
<td>0.18045</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>...</td>
<td>...</td>
<td>...</td>
<td>...</td>
<td>...</td>
</tr>
<tr>
<td>1082</td>
<td>-0.18715</td>
<td>-0.16010</td>
<td>8</td>
<td>9</td>
</tr>
<tr>
<td>1083</td>
<td>-0.18747</td>
<td>-0.15971</td>
<td>8</td>
<td>9</td>
</tr>
<tr>
<td>1084</td>
<td>-0.18747</td>
<td>-0.15951</td>
<td>8</td>
<td>9</td>
</tr>
</tbody>
</table>

Figure 25.4 on page 910 shows the new map after PROC GREMOVE has removed interior state boundaries.

Figure 25.4  Map after Removing Borders (GR25N01(a))

Assign the librefs and set the graphics environment. If the libref MAPS is already assigned, omit the LIBNAME statement.

```sas
libname reflib 'SAS-data-library';
libname maps 'SAS-maps-library';
goptions reset=global gunit=pct border cback=white
    colors=(black blue green red)
    ftext=swiss htitle=6 htext=3;
```

Create data set REFLIB.CBSTATES. This data set includes a variable, DIVISION, that contains the number of the U.S. Bureau of the Census division for the state. This data step converts letter codes to numeric FIPS codes that match those in the STATE variable of MAPS.US.

```sas
data reflib.cbstates;
    length state 8 stcode $ 2 division 4;
```

```sas
```
input stcode division;
state=stfips(stcode);
drop stcode;
datalines;
CT 1
MA 1
...more data lines...
OR 9
WA 9
;

Sort data set in FIPS-code order. Create a sorted data set, CBSORT. It can be properly match-merged with the MAPS.US map data set, which is already sorted in FIPS-code order.

proc sort data=reflib.cbstates out=cbsort;
   by state;
run;

Add DIVISION variable to map data set by merging the CBSORT data set with MAPS.US. Create a new map data set, USCB, that contains all of the state boundary coordinates from the MAPS.US data set plus the added variable DIVISION.

data uscb;
   merge cbsort maps.us;
      by state;
run;

Sort data set in DIVISION order. Sort USCB by the DIVISION variable to create the DIVSTATE data set.

proc sort data=uscb out=divstate;
      by division;
run;

Remove interior boundaries within divisions. BY specifies the variable, DIVISION, in the input map data set that identifies the new unit areas. ID specifies the variable, STATE, in the input map data set that identifies the current unit areas.

proc gremove data=divstate out=remstate;
      by division;
      id state;
run;

Define title and footnote for map.

title 'U.S. State Map';
footnote j=r 'GR25N01(a) ';

Define pattern characteristics.

```
pattern value=mempty repeat=48 color=blue;
```

Show the original map.

```
proc gmap map=maps.us data=maps.us all;
   id state;
   choro state / nolegend;
run;
```

Define new title and footnote for map.

```
title 'U.S. Census Division Map';
footnote j=r 'GR25N01(b) '; 
```

Show the regional map. ID specifies the variable, DIVISION, that identifies the unit areas in the processed data set. CHORO specifies DIVISION as the response variable.

```
proc gmap map=remstate data=remstate all;
   id division;
   choro division / nolegend;
run;
quit;
```

---

**Example 2: Creating an Outline Map of Africa**

**Procedure features:**
- PROC GREMOVE options:
  - DATA=
  - OUT=

**Other features:**
- GMAP procedure

**Sample library member:** GR25N02

This example processes the MAPS.AFRICA map data set, supplied with SAS/GRAPH, to produce a new map data set that contains no internal boundaries. This is done by adding a new variable, REGION, to the map data set and setting it equal to 1. Unit areas from the input map data set that have the same BY-variable value are combined into one unit area in the output map data set. Output 25.3 on page 913 shows the variables present in the original map data set:
Output 25.3  The MAPS.AFRICA Data Set

<table>
<thead>
<tr>
<th>OBS</th>
<th>ID</th>
<th>SEGMENT</th>
<th>X</th>
<th>Y</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>125</td>
<td>1</td>
<td>0.57679</td>
<td>1.43730</td>
</tr>
<tr>
<td>2</td>
<td>125</td>
<td>1</td>
<td>0.57668</td>
<td>1.43467</td>
</tr>
<tr>
<td>3</td>
<td>125</td>
<td>1</td>
<td>0.58515</td>
<td>1.42363</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>...</td>
<td>...</td>
</tr>
<tr>
<td>3462</td>
<td>990</td>
<td>1</td>
<td>1.04249</td>
<td>0.50398</td>
</tr>
<tr>
<td>3463</td>
<td>990</td>
<td>1</td>
<td>1.04184</td>
<td>0.50713</td>
</tr>
<tr>
<td>3464</td>
<td>990</td>
<td>1</td>
<td>1.04286</td>
<td>0.50841</td>
</tr>
</tbody>
</table>

Figure 25.5 on page 913 shows the map before processing:

![Figure 25.5: Map before Removing Borders (GR25N02(a))](image)

The new REFLIB.AFRICA map data set is created with a new variable, REGION. Output 25.4 on page 913 shows the variables that are present in the new map data set created by the GREMOVE procedure:

Output 25.4  The REFLIB.AFRICA Data Set

<table>
<thead>
<tr>
<th>OBS</th>
<th>X</th>
<th>Y</th>
<th>SEGMENT</th>
<th>REGION</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>0.24826</td>
<td>1.02167</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>2</td>
<td>0.25707</td>
<td>1.02714</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>3</td>
<td>0.26553</td>
<td>1.03752</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td></td>
<td>...</td>
<td>...</td>
<td>...</td>
<td>...</td>
</tr>
<tr>
<td>982</td>
<td>1.19071</td>
<td>1.30043</td>
<td>3</td>
<td>1</td>
</tr>
<tr>
<td>983</td>
<td>1.18675</td>
<td>1.30842</td>
<td>3</td>
<td>1</td>
</tr>
<tr>
<td>984</td>
<td>1.18518</td>
<td>1.32822</td>
<td>3</td>
<td>1</td>
</tr>
</tbody>
</table>
Figure 25.6 on page 914 shows the new map after PROC GREMOVE has removed all of the interior boundaries:

![Map after Removing Borders (GR25N02(b))](image)

**Assign the librefs and set the graphics environment.**

```plaintext
libname reflib 'SAS-data-library';
libname maps 'SAS-maps-library';
goptions reset=global gunit=pct border cback=white
   colors=(black blue green red)
   ftext=swiss htitle=6 htext=3;
```

**Create the NEWAF data set.** This new map data set contains all the variables in the SAS/GGRAPH supplied MAPS.AFRICA map data set plus the added variable REGION.

```plaintext
data newaf;
   set maps.africa;
   region=1;
run;
```

**Remove the unit areas from the AFRICA data set.** DATA= specifies the input map data set and OUT= specifies the output map data set. The input map data set has a variable called REGION that is used as the BY-variable to identify the new unit areas. The ID statement specifies the current unit areas from the input map data set.

```plaintext
proc gremove data=newaf out=reflib.africa;
   by region;
   id id;
run;
```
Define the title and footnote.

```plaintext
title 'Africa with Boundaries';
footnote j=r 'GR25N02(a) ';  
```

Define pattern characteristics.

```plaintext
pattern value=mempty r=50 color=blue;
```

Display the original map.

```plaintext
proc gmap map=maps.africa data=maps.africa all;
  id id;
  choro id / nolegend;
run;
```

Define a new title and footnote for the map.

```plaintext
title 'Africa without Boundaries';
footnote j=r 'GR25N02(b) ';  
```

Display the map with no boundaries. ID specifies the variable, REGION, that identifies the unit areas in the processed data set.

```plaintext
proc gmap data=reflib.africa map=reflib.africa;
  id region;
  choro region / nolegend;
run;
quit;
```